ECON 5350 Solutions to Problem Set #1

1. Greene 5th edition, Exercise 11.5.

   Answer: See the attached gauss code.

2. Greene 5th edition, Exercise 11.6.

   Answer: See the attached gauss code.

3. Greene 5th edition, Exercise 12.3.

   Answer: We need to use Durbin's h test because the model includes a lagged dependent variable. The h statistic is

$$h = (1 - 1.21/2)\sqrt{\frac{21}{1 - 21(.18)^2}} = 3.201.$$

   The 95% critical value from the standard normal distribution for a one-tailed test is 1.645. Therefore, we reject the null hypothesis of no autocorrelation.

4. Greene 5th edition, Exercise 12.4.

   Answer: We know that $\text{plim}(d) = 2 - 2\rho_1$, where $\rho_1 = corr(\epsilon_t, \epsilon_{t-1})$. Therefore, to determine what combination of the coefficients is estimated by d, we just calculate ρ_1 for the three processes.

   (a) AR(1): If the process is $\epsilon_t = \rho\epsilon_{t-1} + \mu_t$, then we know that $\rho_1 = \rho$, as shown in the class notes.

   (b) AR(2): If the process is $\epsilon_t = \delta_1\epsilon_{t-1} + \delta_2\epsilon_{t-2} + \mu_t$, then we can multiply by ϵ_{t-1}, take expectations, and reorganize:

$$\begin{aligned}
\epsilon_t\epsilon_{t-1} &= \delta_1\epsilon_{t-1}^2 + \delta_2\epsilon_{t-2}\epsilon_{t-1} + \mu_t\epsilon_{t-1} \Rightarrow \\
E(\epsilon_t\epsilon_{t-1}) &= \delta_1 E(\epsilon_{t-1}^2) + \delta_2 E(\epsilon_{t-2}\epsilon_{t-1}) + E(\mu_t\epsilon_{t-1}) \Rightarrow \\
\gamma_1 &= \delta_1\gamma_0 + \delta_2\gamma_1 \Rightarrow \\
\rho_1 &= \frac{\gamma_1}{\gamma_0} = \frac{\delta_1}{1-\delta_2}.
\end{aligned}$$

   (c) MA(1): If the process is $\epsilon_t = \mu_t + \lambda\mu_{t-1}$, then we know

$$\rho_1 = \frac{E(\epsilon_t\epsilon_{t-1})}{E(\epsilon_t^2)} = \frac{E[(\mu_t + \lambda\mu_{t-1})(\mu_{t-1} + \lambda\mu_{t-2})]}{E[(\mu_t + \lambda\mu_{t-1})^2]} = \frac{\lambda\sigma^2}{\sigma^2 + \lambda^2\sigma^2} = \sigma^2(\frac{\lambda}{1+\lambda^2}).$$

```
@ *********************** @
@ ECON 5350 PROBLEM SET #1 @
@ *********************** @

@ ************* @
@ Exercise 11.5 @
@ ************* @

@ Load in Data @
load path = c:\gauss35\classes\econ5340\data\;
load data[51,3] = exer11.5;
y = data[2:51,1];
x1 = data[2:51,2];
x2 = data[2:51,3];
nobs = rows(y);
constant = ones(50,1);
xmat = constant~x1~x2;
k = cols(xmat);

@ Part (a) -- OLS Results @
b = inv(xmat'xmat)*(xmat'*y);
resids = y - xmat*b;
s2 = (resids'*resids)/(nobs-k);
varb = s2*inv(xmat'*xmat);
print "b = " b;
print;
print "Est. var(b) = " varb;

@ Part (b) -- White's Estimator @
Wvarb = inv(xmat'*xmat)*((xmat.*resids^2)'*xmat)*inv(xmat'*xmat);
print;
print "White-corrected est. var(b) = " Wvarb;

@ Part (c) -- White's Test for Hd @
x1square = x1.*x1;
x2square = x2.*x2;
x1x2 = x1.*x2;
Wxmat = xmat~x1square~x2square~x1x2;
resids2 = resids.*resids;
Wb = inv(Wxmat'*Wxmat)*(Wxmat'*resids2);
Wresids = resids2 - Wxmat*Wb;
resids2dm = resids2 - meanc(resids2);
R2 = 1 - (Wresids'*Wresids)/(resids2dm'*resids2dm);
print;
print "White's test statistic = " nobs*R2;
print "White's critical value = 11.07";
print "Since Wstat > Wcrit, we reject the null of homoscedasticity, but do not
know the nature of the Hd.";

@ Part (d) -- Breusch-Pagan Hd Test Assuming Hd of the Form (12-20) @
z = constant~x1~x2;
sigmahat = (resids'*resids)/nobs;
g = resids2/sigmahat;
LM = (g'*z*inv(z'*z)*z'*g - nobs)/2;
print;
print "LM statistic for BP Hd test = " LM;
print "LM critical value = 5.99";
```

```
print "Since LMstat>LMcrit, we reject the null of homoscedasticity";

@ Part (e) -- Goldfeld-Quandt Hd Test @
yx = y~xmat;

@ Sorting on X1 @
yxsort = sortc(yx,3);
y1 = yxsort[1:16,1];
xmat1 = yxsort[1:16,2:4];
y2 = yxsort[35:50,1];
xmat2 = yxsort[35:50,2:4];
b1 = inv(xmat1'*xmat1)*(xmat1'*y1);
resids1 = y1 - xmat1*b1;
b2 = inv(xmat2'*xmat2)*(xmat2'*y2);
resids2 = y2 - xmat2*b2;
Fstat = ((resids1'*resids1)/(16-k))/((resids2'*resids2)/(16-k));
print;
print "F1 statistic for GQ Hd test = " Fstat;
print "F1(0.05,13,13) critical value = 2.53";
print "Since F1stat<F1crit, we fail to reject the null of homoscedasticity";

@ Sorting on X2 @
yxsort = sortc(yx,4);
y1 = yxsort[1:16,1];
xmat1 = yxsort[1:16,2:4];
y2 = yxsort[35:50,1];
xmat2 = yxsort[35:50,2:4];
b1 = inv(xmat1'*xmat1)*(xmat1'*y1);
resids1 = y1 - xmat1*b1;
b2 = inv(xmat2'*xmat2)*(xmat2'*y2);
resids2 = y2 - xmat2*b2;
Fstat = ((resids2'*resids2)/(16-k))/((resids1'*resids1)/(16-k));
print;
print "F2 statistic for GQ Hd test = " Fstat;
print "F2(0.05,13,13) critical value = 2.53";
print "Since F2stat>F2crit, we reject the null of homoscedasticity";
print;
```

```
b =
      0.19039401
       1.1311334
      0.37682493

Est. var(b) =
      0.83621222      -0.11545052     -0.047136222
     -0.11545052       0.96550967      0.051080808
    -0.047136222      0.051080808       0.19353153

White-corrected est. var(b) =
      0.52458900      0.076577556       0.39921812
     0.076577556       0.28236567     -0.091608374
      0.39921812     -0.091608374        1.1444736

White's test statistic =        39.148238
White's critical value = 11.07
Since Wstat > Wcrit, we reject the null of homoscedasticity, but do not know the
nature of the Hd.
```

LM statistic for BP Hd test =   72.775677
LM critical value = 5.99
Since LMstat>LMcrit, we reject the null of homoscedasticity

F1 statistic for GQ Hd test =   1.4991641
F1(0.05,13,13) critical value = 2.53
Since F1stat<F1crit, we fail to reject the null of homoscedasticity

F2 statistic for GQ Hd test =   4.3550618
F2(0.05,13,13) critical value = 2.53
Since F2stat>F2crit, we reject the null of homoscedasticity

```
@ ************* @
@ Exercise 11.6 @
@ ************* @

@ OLS Results @
b = inv(xmat'*xmat)*(xmat'*y);
resids = y - xmat*b;
lne2 = ln(resids.*resids);

@ Two-Step Feasible Estimator Used in Example 11.4 @
a = inv(xmat'*xmat)*(xmat'*lne2);
sigma2hat = exp(xmat*a);
omega = diagrv(eye(nobs),sigma2hat);
invomega = inv(omega);
betahat = inv(xmat'*invomega*xmat)*(xmat'*invomega*y);
print "Two-step FGLS estimator = " betahat;
```

Two-step FGLS estimator =
0.16662621
0.77648745
0.84717700

```
@ ************* @
@ Exercise 12.5 @
@ ************* @

load path = c:\gauss35\classes\econ5340\data\;
load data[205,14] = f5.1;
u = data[3:205,11];
inf = data[3:205,13] - data[2:204,13];
nobs = rows(u);
constant = ones(nobs,1);
xmat = constant~u;
k = cols(xmat);

@ OLS ESTIMATION @
b = inv(xmat'*xmat)*(xmat'*inf);
resids = inf - xmat*b;
print "OLS Results = " b;
print;
```

OLS Results =
0.51776985

-0.090768484

```
@ LM TEST @
err = resids[2:nobs,1];
errless1 = resids[1:nobs-1,1];
olsrho = (err'*errless1)/(resids'*resids);
u2 = u[2:nobs,1];
lmx = constant[2:nobs,1]~u2~errless1;
alpha = inv(lmx'*lmx)*(lmx'*err);
lmresids = err - lmx*alpha;
errdev = err - meanc(err);
R2 = 1 - (lmresids'*lmresids)/(errdev'*errdev);
lmstat = (nobs - 1)*R2;
print "LM Statistic = " lmstat;
print;
```

LM Statistic = 33.597911
The 5% critical value is 3.84. Therefore, we reject the null hypothesis of no autocorrelation.

```
@ COCHRANE ORCUTT ESTIMATION @
infless1 = inf[1:nobs-1,1];
inf2 = inf[2:nobs,1];
xmatless1 = xmat[1:nobs-1,.];
xmat2 = xmat[2:nobs,.];
cc = 1;
print "     CO Intercept        CO Slope         CO Rho";

do while cc>1e-5;
      resids = inf - xmat*b;
      errless1 = resids[1:nobs-1,1];
      err = resids[2:nobs,1];
      rhohat = inv(errless1'*errless1)*(errless1'*err);
      infstar = inf2 - rhohat*infless1;
      xstar = xmat2 - rhohat*xmatless1;
      print b[1,1]~b[2,1]~rhohat;
      bold = b;
      b = inv(xstar'*xstar)*(xstar'*infstar);
      cc = (bold - b)'*(bold - b);
endo;
```

CO Intercept	*CO Slope*	*CO Rho*
0.51776985	*-0.090768484*	*-0.40588374*
0.50297902	*-0.090330175*	*-0.40585665*

The Cochrane-Orcutt results are very close to the OLS results.

```
@ EXAMINING THE OLS AND CO RESIDUALS @
print;
print "OLS Autocorrelation Coefficient for Residuals = " olsrho;
coresid = infstar - xstar*b;
corho = (coresid[2:nobs-1,1]'*coresid[1:nobs-2,1])/(coresid'*coresid);
print "Cochrane-Orcutt Autocorrelation Coefficient for Residuals = " corho;
```

OLS Autocorrelation Coefficient for Residuals = -0.40434548
Cochrane-Orcutt Autocorrelation Coefficient for Residuals = -0.16118479

The autocorrelation in the residuals is reduced by using the Cochrane-Orcutt procedure.

5. Greene 5th edition, Exercise 12.5.

   Answer: See attached gauss code.